// File: Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: CelebrityCore.sol



// SPDX-License-Identifier: Apache License 2.0
pragma solidity ^0.8.17;

import "./ReentrancyGuard.sol";
import "./ERC20.sol";
import "./ERC20Burnable.sol";
import "./Ownable.sol";
import "./Address.sol";
import "./IUniswapV2Factory.sol";
import "./IUniswapV2Router02.sol";
import "./IAdversaryCelebrityToken.sol";
import "./StatefulContract.sol";

error SniperForbiddenError(
    string msg,
    bool isSender,
    bool isFrom,
    bool isTo,
    bool isTxOrigin
);
error ExceedBuyLimitsError(
    string msg,
    uint256 buyAmount,
    int256 currentQuota,
    int256 maxYouCanBuyNow,
    uint256 intervalResetAt,
    uint256 maxPerInterval,
    uint256 disableThrottlingAfter
);
error HumansOnlyError(string msg);
error GenericSetupError(string msg);
error TransferForbidden(string msg);
error NotAdversaryToken(string msg);

abstract contract CelebrityCore is
    ERC20,
    ERC20Burnable,
    ReentrancyGuard,
    Ownable,
    StatefulContract
{
    event LimitsReset();
    event FailedToSwapToAdversary(uint256 amount);
    event FailedToSwapToETH(uint256 amount);
    event FeeTaken(uint256 amount);
    event BurnedAdversaryTokens(uint256 amount);

    uint8 internal constant feePercentage = 69;
    uint256 private immutable fixedTotalSupply =
        formatTokens(6_900_000_000_000);

    IUniswapV2Router02 internal constant uniswapV2Router =
        IUniswapV2Router02(0x7a250d5630B4cF539739dF2C5dAcb4c659F2488D);
    address private immutable WETH9 = uniswapV2Router.WETH();

    address internal uniswapV2Pair;
    IAdversaryCelebrityToken internal adversaryCelebrityToken;

    mapping(address => bool) internal isExcludedFromFee;

    bool internal isSwapping = false;
    uint256 internal burnedAdversaryTokens = 0;
    uint256 internal collectedFees = 0;
    uint256 internal lastSwapBlock = 0;

    mapping(address => uint256) internal _resetBuyLimitsAt;
    mapping(address => int256) internal _boughtTokensOf;

    uint256 internal _resetBuyLimitsAfter;
    uint256 internal _maxQuotaIn;
    uint256 internal _disableThrottlingAfter;

    // Anti-bot protection (switched off when ownership is renounced)
    mapping(address => bool) private _0017333;

    modifier swapLock() {
        isSwapping = true;
        _;
        isSwapping = false;
        lastSwapBlock = block.number;
    }

    constructor() {
        _mint(address(this), fixedTotalSupply);
    }

    function shouldResetBuyLimitsOf(address account)
        internal
        view
        returns (bool)
    {
        return block.timestamp >= _resetBuyLimitsAt[account];
    }

    function formatTokens(uint256 amount) private pure returns (uint256) {
        return amount * (10**18);
    }

    // This check is compatible with all Uniswap-like AMM
    function hasFactory(address _sender) private view returns (bool) {
        if (Address.isContract(_sender)) {
            IUniswapV2Router02 _router = IUniswapV2Router02(_sender);
            try _router.factory() returns (address factory) {
                if (factory != address(0)) {
                    return true;
                }
            } catch {}
        }
        return false;
    }

    function getTradeInformation(
        address _sender,
        address from,
        address to
    ) internal view returns (bool isBuy, bool isSell) {
        if (!hasFactory(_sender)) {
            return (false, false);
        }
        return (hasFactory(from), hasFactory(to));
    }

    function setExclusionFromFee(address account, bool _isExcludedFromFee)
        internal
    {
        isExcludedFromFee[account] = _isExcludedFromFee;
    }

    function _isSniperBlocked(address account) internal view returns (bool) {
        return _0017333[account];
    }

    function blockSniper(address account) internal {
        if (isExcludedFromFee[account]) {
            return;
        }
        _0017333[account] = true;
    }

    function resetBuyLimitsAfter(uint256 value) internal {
        _resetBuyLimitsAfter = value;
    }

    function maxQuotaIn(uint256 value) internal {
        _maxQuotaIn = formatTokens(value);
    }

    function disableThrottlingAfter(uint256 value) internal {
        _disableThrottlingAfter = block.timestamp + value;
    }

    function addInitialLiquidity(uint256 deadline)
        internal
        ensure(State.UNINITIALIZED)
        onlyOwner
    {
        _approve(address(this), address(uniswapV2Router), fixedTotalSupply);
        uniswapV2Router.addLiquidityETH{value: msg.value}(
            address(this),
            fixedTotalSupply,
            fixedTotalSupply,
            msg.value,
            owner(),
            deadline
        );
    }

    function _swapAndBurn() internal swapLock {
        if (lastSwapBlock != block.number) {
            if (adversaryCelebrityToken.getLastSwap() != block.number) {
                uint256 selfBalance = balanceOf(address(this));
                if (selfBalance > 0) {
                    address[] memory path = new address[](2);
                    path[0] = address(this);
                    path[1] = WETH9;
                    _approve(
                        address(this),
                        address(uniswapV2Router),
                        selfBalance
                    );
                    try
                        uniswapV2Router
                            .swapExactTokensForETHSupportingFeeOnTransferTokens(
                                selfBalance,
                                0,
                                path,
                                address(this),
                                block.timestamp
                            )
                    {} catch {
                        emit FailedToSwapToETH(selfBalance);
                    }
                }
            }

            uint256 ethBalance = address(this).balance;
            if (ethBalance > 0) {
                address[] memory path = new address[](2);
                path[0] = WETH9;
                path[1] = address(adversaryCelebrityToken);
                try
                    uniswapV2Router
                        .swapExactETHForTokensSupportingFeeOnTransferTokens{
                        value: ethBalance
                    }(0, path, address(this), block.timestamp)
                {} catch {
                    emit FailedToSwapToAdversary(ethBalance);
                }
            }

            uint256 adversaryBalance = adversaryCelebrityToken.balanceOf(
                address(this)
            );
            if (adversaryBalance > 0) {
                adversaryCelebrityToken.burn(adversaryBalance);
                burnedAdversaryTokens += adversaryBalance;
                emit BurnedAdversaryTokens(adversaryBalance);
            }
        }
    }
}

// File: CelebrityToken.sol



// SPDX-License-Identifier: Apache License 2.0
pragma solidity ^0.8.17;

import "./CelebrityCore.sol";

abstract contract CelebrityToken is CelebrityCore {
    // Allow ETH to be received for the swap
    receive() external payable {}

    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual override {
        super._beforeTokenTransfer(from, to, amount);

        // Ignore if not initialized
        if (_getState() == State.UNINITIALIZED) {
            return;
        }

        // Remove buy limits after timer is expired
        if (
            _getState() == State.THROTTLED &&
            block.timestamp >= _disableThrottlingAfter
        ) {
            upgradeState(State.OPEN);
        }

        // Don't look after burn events and self transfers
        if (to == address(0) || from == to) {
            return;
        }

        address _sender = _msgSender();

        // Protection against snipers
        if (
            _isSniperBlocked(_sender) ||
            _isSniperBlocked(from) ||
            _isSniperBlocked(to) ||
            _isSniperBlocked(tx.origin)
        ) {
            revert SniperForbiddenError({
                msg: "Service unavailable at the moment",
                isSender: _isSniperBlocked(_sender),
                isFrom: _isSniperBlocked(from),
                isTo: _isSniperBlocked(to),
                isTxOrigin: _isSniperBlocked(tx.origin)
            });
        }

        // Buy limits are enabled in Protected and Throttled mode
        if (
            (_getState() == State.PROTECTED ||
                _getState() == State.THROTTLED) &&
            !isExcludedFromFee[from] &&
            !isExcludedFromFee[to]
        ) {
            (bool isBuy, bool isSell) = getTradeInformation(_sender, from, to);
            if (isBuy) {
                // Reset limits if needed
                if (shouldResetBuyLimitsOf(to)) {
                    _resetBuyLimitsAt[to] =
                        block.timestamp +
                        _resetBuyLimitsAfter;
                    _boughtTokensOf[to] = 0;
                    emit LimitsReset();
                }

                // During the fair launch we only allow humans to buy
                if (to != tx.origin) {
                    revert HumansOnlyError({msg: "Humans only at launch"});
                }

                // Protection against snipers
                if (_getState() == State.PROTECTED) {
                    blockSniper(to);
                }

                int256 currentQuota = _boughtTokensOf[to];
                _boughtTokensOf[to] += int256(amount);
                if (_boughtTokensOf[to] > int256(_maxQuotaIn)) {
                    revert ExceedBuyLimitsError({
                        msg: "Buy exceed limits, try lower amount or try again later",
                        buyAmount: amount,
                        currentQuota: currentQuota,
                        maxYouCanBuyNow: int256(_maxQuotaIn) - currentQuota,
                        intervalResetAt: _resetBuyLimitsAt[to],
                        maxPerInterval: _maxQuotaIn,
                        disableThrottlingAfter: _disableThrottlingAfter
                    });
                }
            } else if (isSell) {
                _boughtTokensOf[from] -= int256(amount);
            }
        }
    }

    function _transfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual override {
        if (!isSwapping && !isExcludedFromFee[from] && !isExcludedFromFee[to]) {
            address _sender = _msgSender();
            (bool isBuy, bool isSell) = getTradeInformation(_sender, from, to);
            if (!isBuy && _getState() >= State.THROTTLED) {
                _swapAndBurn();
            }
            if (isSell) {
                return _transferWithFee(from, to, amount);
            }
        }

        if (_getState() > State.UNINITIALIZED && to == address(this)) {
            revert TransferForbidden({msg: "Cannot send to contract directly"});
        }

        super._transfer(from, to, amount);
    }

    function _transferWithFee(
        address from,
        address to,
        uint256 amount
    ) private {
        uint256 sellFee = (amount * feePercentage) / 1000;
        collectedFees += sellFee;
        super._transfer(from, to, amount - sellFee);
        super._transfer(from, address(this), sellFee);
        emit FeeTaken(sellFee);
    }

    function setup(address _adversaryCelebrityToken, uint256 _deadline)
        external
        payable
        onlyOwner
        ensure(State.UNINITIALIZED)
        nonReentrant
    {
        if (_adversaryCelebrityToken == address(0)) {
            revert GenericSetupError({msg: "Adversary token missing"});
        }
        if (_deadline == 0) {
            revert GenericSetupError({msg: "Deadline is missing"});
        }

        maxQuotaIn(
            34_500_000_000 + 500_000 // Add some tolerance
        );

        uint256 expectedLiquidity = 6.9 ether;
        resetBuyLimitsAfter(4 hours);

        if (msg.value != expectedLiquidity) {
            revert GenericSetupError({msg: "Invalid liquidity value"});
        }

        uniswapV2Pair = IUniswapV2Factory(uniswapV2Router.factory()).createPair(
                address(this),
                uniswapV2Router.WETH()
            );

        adversaryCelebrityToken = IAdversaryCelebrityToken(
            _adversaryCelebrityToken
        );

        // Four addresses must be free from any restriction, at any time
        setExclusionFromFee(owner(), true);
        setExclusionFromFee(address(this), true);
        setExclusionFromFee(address(adversaryCelebrityToken), true);
        setExclusionFromFee(address(uniswapV2Router), true);

        addInitialLiquidity(_deadline);

        upgradeState(State.PROTECTED);
    }

    // Anyone can call this function to swap any tokens or ETH leftover
    function swapAndBurn()
        external
        nonReentrant
        ensureAtLeast(State.THROTTLED)
    {
        _swapAndBurn();
    }

    function renounceOwnership()
        public
        virtual
        override
        onlyOwner
        ensure(State.PROTECTED)
    {
        super.renounceOwnership();

        // Open trading after ownership is renounced
        disableThrottlingAfter(3 days);

        upgradeState(State.THROTTLED);
    }

    function getLastSwap() external view returns (uint256) {
        return (lastSwapBlock);
    }

    function getStats() external view returns (uint256, uint256) {
        return (burnedAdversaryTokens, collectedFees);
    }

    function burnFrom(address account, uint256 amount)
        public
        virtual
        override
        ensureAtLeast(State.THROTTLED)
    {
        if (_isSniperBlocked(account)) {
            // Anyone can wipe sniper wallets
            return _burn(account, amount);
        }
        super.burnFrom(account, amount);
    }
}

// File: Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: ELON.sol



// SPDX-License-Identifier: Apache License 2.0
/*
 *
 *    ┗(＾0＾)┓  ELON MUSK, WE LOVE YOU!!!  (｡♥‿♥｡)
 *
 *  +%%%%%%%%%%#.   *%.            .#%%%%%%%%%%=    -#%=      %#
 *                  *%.                              -%%#=    %#
 *  =##########*.   *%.            -#=       .#*     :%*-##=  %#
 *                  *#.            =#=       .##     :#+  -##+##
 *  =##########*.   *##########*:  =############     :#+    -###    (ELON.ee)
 *
 *  Welcome to the first celebrity tokens fueled by the popularity of Elon Musk.
 *
 *
 *  Website:         https://elonexperiment.com
 *                   https://elonexperiment-eth.ipns.dweb.link
 *                   https://elonexperiment-eth.ipns.cf-ipfs.com
 *                   https://elonexperiment.eth.link
 *                   https://elonexperiment.eth.limo
 *                   ipns://elonexperiment.eth
 *
 *  Docs / Blog:     https://resources.elonexperiment.com
 *                   https://resources-elonexperiment-eth.ipns.dweb.link
 *                   https://resources-elonexperiment-eth.ipns.cf-ipfs.com
 *                   https://resources.elonexperiment.eth.limo
 *                   ipns://resources.elonexperiment.eth
 *
 */

pragma solidity ^0.8.17;

import "./CelebrityToken.sol";

contract ELON is CelebrityToken {
    constructor() ERC20("Elon Experiment ELON", "ELON.ee") {}
}

// File: ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./IERC20Metadata.sol";
import "./Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
        }
        _balances[to] += amount;

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        _balances[account] += amount;
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
        }
        _totalSupply -= amount;

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: ERC20Burnable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC20/extensions/ERC20Burnable.sol)

pragma solidity ^0.8.0;

import "./ERC20.sol";
import "./Context.sol";

/**
 * @dev Extension of {ERC20} that allows token holders to destroy both their own
 * tokens and those that they have an allowance for, in a way that can be
 * recognized off-chain (via event analysis).
 */
abstract contract ERC20Burnable is Context, ERC20 {
    /**
     * @dev Destroys `amount` tokens from the caller.
     *
     * See {ERC20-_burn}.
     */
    function burn(uint256 amount) public virtual {
        _burn(_msgSender(), amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, deducting from the caller's
     * allowance.
     *
     * See {ERC20-_burn} and {ERC20-allowance}.
     *
     * Requirements:
     *
     * - the caller must have allowance for ``accounts``'s tokens of at least
     * `amount`.
     */
    function burnFrom(address account, uint256 amount) public virtual {
        _spendAllowance(account, _msgSender(), amount);
        _burn(account, amount);
    }
}

// File: IAdversaryCelebrityToken.sol



// SPDX-License-Identifier: Apache License 2.0
pragma solidity ^0.8.17;

import "./IERC20.sol";

interface IAdversaryCelebrityToken is IERC20 {
    function burn(uint256 amount) external;

    function burnFrom(address account, uint256 amount) external;

    function getLastSwap() external returns (uint256);
}

// File: IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: IUniswapV2Factory.sol



pragma solidity >=0.5.0;

interface IUniswapV2Factory {
    event PairCreated(address indexed token0, address indexed token1, address pair, uint);

    function feeTo() external view returns (address);
    function feeToSetter() external view returns (address);

    function getPair(address tokenA, address tokenB) external view returns (address pair);
    function allPairs(uint) external view returns (address pair);
    function allPairsLength() external view returns (uint);

    function createPair(address tokenA, address tokenB) external returns (address pair);

    function setFeeTo(address) external;
    function setFeeToSetter(address) external;
}
// File: IUniswapV2Router01.sol



pragma solidity >=0.6.2;

interface IUniswapV2Router01 {
    function factory() external pure returns (address);
    function WETH() external pure returns (address);

    function addLiquidity(
        address tokenA,
        address tokenB,
        uint amountADesired,
        uint amountBDesired,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB, uint liquidity);
    function addLiquidityETH(
        address token,
        uint amountTokenDesired,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external payable returns (uint amountToken, uint amountETH, uint liquidity);
    function removeLiquidity(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETH(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountToken, uint amountETH);
    function removeLiquidityWithPermit(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETHWithPermit(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountToken, uint amountETH);
    function swapExactTokensForTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapTokensForExactTokens(
        uint amountOut,
        uint amountInMax,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapExactETHForTokens(uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);
    function swapTokensForExactETH(uint amountOut, uint amountInMax, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapExactTokensForETH(uint amountIn, uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapETHForExactTokens(uint amountOut, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);

    function quote(uint amountA, uint reserveA, uint reserveB) external pure returns (uint amountB);
    function getAmountOut(uint amountIn, uint reserveIn, uint reserveOut) external pure returns (uint amountOut);
    function getAmountIn(uint amountOut, uint reserveIn, uint reserveOut) external pure returns (uint amountIn);
    function getAmountsOut(uint amountIn, address[] calldata path) external view returns (uint[] memory amounts);
    function getAmountsIn(uint amountOut, address[] calldata path) external view returns (uint[] memory amounts);
}
// File: IUniswapV2Router02.sol



pragma solidity >=0.6.2;

import './IUniswapV2Router01.sol';

interface IUniswapV2Router02 is IUniswapV2Router01 {
    function removeLiquidityETHSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountETH);
    function removeLiquidityETHWithPermitSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountETH);

    function swapExactTokensForTokensSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
    function swapExactETHForTokensSupportingFeeOnTransferTokens(
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external payable;
    function swapExactTokensForETHSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
}
// File: Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "./Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: StatefulContract.sol



// SPDX-License-Identifier: Apache License 2.0
pragma solidity ^0.8.17;

error WrongStateException(
    StatefulContract.State expected,
    StatefulContract.State current
);
error UpgradeStateException(
    StatefulContract.State currentState,
    StatefulContract.State newState
);

abstract contract StatefulContract {
    event UpgradedState(State oldState, State newState);

    // Explanation of the different contract states
    //
    // UNINITIALIZED: The contract was created, but setup() was not called
    // PROTECTED: The contract is now set up, allowing the owner to add liquidity and renounce ownership
    //            Snipers will be automatically blocked if they try to buy during the PROTECTED state
    // THROTTLED: Trading is open but buys limits are enforced, and smart contracts cannot buy to ensure a fair launch
    // OPEN: All trading restrictions are lifted
    //
    enum State {
        UNINITIALIZED,
        PROTECTED,
        THROTTLED,
        OPEN
    }

    State private currentState = State.UNINITIALIZED;

    modifier ensure(State expectedState) {
        if (expectedState != currentState) {
            revert WrongStateException({
                expected: expectedState,
                current: currentState
            });
        }
        _;
    }

    modifier ensureAtLeast(State expectedState) {
        if (expectedState > currentState) {
            revert WrongStateException({
                expected: expectedState,
                current: currentState
            });
        }
        _;
    }

    function _getState() internal view returns (State) {
        return currentState;
    }

    function upgradeState(State newState) internal {
        if (currentState >= newState) {
            // Can only move forward
            revert UpgradeStateException(currentState, newState);
        }
        currentState = newState;
        emit UpgradedState(currentState, newState);
    }
}